Exploration Geophysics (1986) **17**, 189–200

MRF/375

Magnetic and gravity anomalies of a triaxial ellipsoid

D. A. Clark,[1] S. J. Saul[2] and D. W. Emerson[3]

[1]*CSIRO Division of Mineral Physics and Mineralogy, PO Box 136, North Ryde, NSW 2113, Australia*
[2]*Austirex International Ltd, 27 Merriwa Street, Gordon, NSW 2072, Australia*
[3]*Department of Geology and Geophysics, University of Sydney, NSW 2006, Australia*

Abstract

The theoretical background, a computational algorithm and an HP41CX calculator program for forward modelling of magnetic and gravity anomalies due to ellipsoidal bodies are presented, including gravity effects of prolate and oblate ellipsoids of revolution and two-dimensional elliptic cylinders. The ellipsoid is particularly useful for modelling strongly magnetic, compact orebodies, because of the flexibility and appropriateness of the geometric form and, in particular, because ellipsoids are the only bodies for which self-demagnetization can be treated exactly and analytically. Remanence, anisotropic susceptibility and self-demagnetization are all included in the analysis.

Key words: ellipsoidal co-ordinates, elliptic integrals, gravity anomalies, magnetic anomalies, potential theory, remanence, self-demagnetization, susceptibility anisotropy.

Introduction

A great many geometric forms have been used for modelling potential field anomalies in the space domain. Recently, Emerson *et al* (1985) presented formulae and hand-held calculator algorithms for calculating magnetic anomalies due to a variety of commonly used models. The aim of that publication was to compile an accessible, readily comprehensible and relatively comprehensive suite of models, employing consistent notation throughout, which incorporated remanence, anisotropy and (at least approximately) self-demagnetization. This compilation was, for the most part, based on previous work, scattered throughout the literature, which was checked for errors and corrected if necessary.

A significant omission from previous publications has been the ellipsoid model, in spite of its theoretical attractiveness and extensive practical experience with a proprietary algorithm that has demonstrated its utility (Farrar 1979). The ellipsoid model has several advantages:

(1) It is the only model that allows self-demagnetization to be taken into account analytically. This feature is important for bodies of high susceptibility, including many magnetic orebodies.

(2) The triaxial ellipsoid is a very flexible model that provides a good representation of a variety of discrete sources, both compact and elongated, with prolate and oblate ellipsoids of revolution, the two-dimensional (2D) elliptic cylinder and the sphere as special cases.

(3) Many orebodies are found to conform approximately to an ellipsoidal form. Homogeneous deformation of an originally compact body produces an ellipsoidal shape, as for the Precambrian, stratabound, sulphide orebodies of Finland, which take the form of highly elongated triaxial ellipsoids with major axes parallel to the tectonic lineation and a,b planes parallel to the axial plane cleavage (Gaal 1977). Further examples include the massive magnetite bodies ('ironstones') of the Tennant Creek field in the Northern Territory (Farrar 1979) and the sulphide orebodies of the Cobar district, NSW, including the Elura deposit, which both take the form of more or less elongated ellipsoids flattened in the plane of the cleavage. The shape of these bodies may reflect a syntectonic mode of deposition (Solomon *et al* 1986).

Although the ellipsoid model appears to have been independently developed and used by various workers, it has evidently been considered too valuable for general dissemination. Accordingly, only sketchy details have ever been published (e.g. Hjelt & Turunen 1981) and the exploration profession has hitherto been deprived of an exceptionally useful interpretational tool. Recently, Pedersen (1985) has presented expressions for the gravity and magnetic effects of ellipsoidal bodies in the wavenumber (spatial frequency) domain. Presented here are some basic theory and the expressions for gravity and magnetic anomalies as functions of spatial co-ordinates. The formulae are set out in a step-by-step format with numerous sub-headings to facilitate programming.

Ellipsoidal co-ordinates

The equation of the surface of an ellipsoid with semi-axes $a > b > c$ is

$$(x_1^2/a^2)+(x_2^2/b^2)+(x_3^2/c^2)=1 \qquad (1)$$

where x_1, x_2, x_3 are Cartesian co-ordinates with respect to the principal axes (body axes), with the origin of co-ordinates at the centre of the ellipsoid (Stratton 1941).

The foci of the ellipsoid are at the points on the principal axes

$$x_1=\pm\sqrt{(a^2-b^2)},\ x_2=x_3=0$$
$$x_1=\pm\sqrt{(a^2-c^2)},\ x_2=x_3=0$$
$$x_2=\pm\sqrt{(b^2-c^2)},\ x_1=x_3=0$$

Received 1 September 1986; accepted for publication 1 November 1986.

The equation

$$x_1^2/(a^2+\lambda)+x_2^2/(b^2+\lambda)+x_3^2/(c^2+\lambda)=1 \qquad (\lambda>-c^2) \quad (2)$$

defines a family of ellipsoidal surfaces and by the preceding formulae it is apparent that they are confocal with the basic ellipsoid defined by equation (1), for which $\lambda=0$. As λ approaches ∞ the equation of the confocal ellipsoidal surface becomes $(x_1^2/\lambda)+(x_2^2/\lambda)+(x_3^2/\lambda)=1$, or $x_1^2+x_2^2+x_3^2=\lambda$. This is the equation of a sphere with radius $r=\sqrt{\lambda}$, so as λ increases without bound the confocal surfaces become spherical.

Points external to the basic ellipsoid correspond to $\lambda>0$, internal points to $\lambda<0$, and through any point there is only one such ellipsoidal surface. Specification of a point in space requires two further families of intersecting surfaces: the hyperboloids of one sheet characterized by the parameter μ, where

$$x_1^2/(a^2+\mu)+x_2^2/(b^2+\mu)+x_3^2/(c^2+\mu)=1 \qquad (-c^2>\mu>-b^2); \quad (3)$$

and the hyperboloids of two sheets, characterized by the parameter ν,

$$x_1^2/(a^2+\nu)+x_2^2/(b^2+\nu)+x_3^2/(c^2+\nu)=1 \qquad (-b^2>\nu>-a^2). \quad (4)$$

It can be shown that the three sets of confocal surfaces are mutually orthogonal. A point (x_1, x_2, x_3) which lies at the intersection of three surfaces corresponding to parameters λ, μ, ν is said to have ellipsoidal co-ordinates (λ, μ, ν). These ellipsoidal co-ordinates can be found as the largest, intermediate and smallest roots respectively of the cubic equation:

$$x_1^2/(a^2+s)+x_2^2/(b^2+s)+x_3^2/(c^2+s)=1 \qquad (s=\lambda,\ \mu,\ \nu) \quad (5)$$

or

$$s^3+p_2s^2+p_1s+p_0=0 \quad (6)$$

where

$$p_2=a^2+b^2+c^2-x_1^2-x_2^2-x_3^2$$

$$p_1=a^2b^2+b^2c^2+c^2a^2-(b^2+c^2)x_1^2-(c^2+a^2)x_2^2-(a^2+b^2)x_3^2$$

$$p_0=a^2b^2c^2-b^2c^2x_1^2-c^2a^2x_2^2-a^2b^2x_3^2.$$

Equation (6) has three real roots. In descending order they are:

$$\lambda=\ 2\sqrt{(-p/3)}\ \cos(\theta/3)-p_2/3 \quad (7)$$

$$\mu=\ -2\sqrt{(-p/3)}\ \cos(\theta/3+\pi/3)-p_2/3 \quad (8)$$

$$\nu=\ -2\sqrt{(-p/3)}\ \cos(\theta/3-\pi/3)-p_2/3, \quad (9)$$

where

$$\theta=\cos^{-1}[-q/2\sqrt{(-p/3)^3}],$$

$$p=p_1-p_2^2/3,$$

$$q=p_0-p_1p_2/3+2(p_2/3)^3.$$

Expressions for fields due to an ellipsoid involve the spatial derivatives of λ, viz. $\partial\lambda/\partial x_j$ ($j=1, 2, 3$). These are obtained by differentiating equation (2). Thus x_1 – differentiation yields:

$$\frac{2x_1}{(a^2+\lambda)}-\frac{x_1^2}{(a^2+\lambda)^2}\frac{\partial\lambda}{\partial x_1}-\frac{x_2^2}{(b^2+\lambda)^2}\frac{\partial\lambda}{\partial x_1}-\frac{x_3^2}{(c^2+\lambda)^2}\frac{\partial\lambda}{\partial x_1}=0$$

Rearranging, we get:

$$\frac{\partial\lambda}{\partial x_1}=\frac{2x_1/(a^2+\lambda)}{\left(\frac{x_1}{a^2+\lambda}\right)^2\left(\frac{x_2}{b^2+\lambda}\right)^2\left(\frac{x_3}{c^2+\lambda}\right)^2} \quad (10)$$

The other spatial derivatives are easily written down by symmetry.

The ellipsoid problem of potential theory

Calculating the gravitational potential of a homogeneous ellipsoid is a classical problem of potential theory, which was formally solved in 1839 by Dirichlet. The external potential U_e of an ellipsoid with density ϱ and semi-axes a ≥ b ≥ c is given by the following expression (Kellogg 1929):

$$U_e\ (x_1,x_2,x_3)=\pi abcG\varrho\int_\lambda^\infty\left[1-\frac{x_1^2}{a^2+u}-\frac{x_2^2}{b^2+u}-\frac{x_3^2}{c^2+u}\right]\frac{du}{R(u)} \quad (11)$$

where G is the gravitational constant and

$$R(u)=[\ (a^2+u)(b^2+u)(c^2+u)\ ]^{0.5}. \quad (12)$$

Inside the ellipsoid the potential is given by:

$$U_i\ (x_1,x_2,x_3)=\pi abcG\varrho\int_0^\infty\left[1-\frac{x_1^2}{a^2+u}-\frac{x_2^2}{b^2+u}-\frac{x_3^2}{c^2+u}\right]\frac{du}{R(u)} \quad (13)$$

Note that U_i involves only terms of the form: (constant) or (constant × x_i^2), so that within the ellipsoid the gravitational field components $\partial U_i/\partial x_j$ ($j=1, 2, 3$) are linear in the Cartesian co-ordinates x_1, x_2, x_3. Equation (11) may be rewritten:

$$U_e=\pi abcG\varrho[D(\lambda)-A(\lambda)x_1^2-B(\lambda)x_2^2-C(\lambda)x_3^2], \quad (14)$$

where

$$D(\lambda)=\int_\lambda^\infty\frac{du}{R(u)} \quad (15)$$

$$A(\lambda)=\int_\lambda^\infty\frac{du}{(a^2+u)R(u)} \quad (16)$$

$$B(\lambda)=\int_\lambda^\infty\frac{du}{(b^2+u)R(u)} \quad (17)$$

$$C(\lambda)=\int_\lambda^\infty\frac{du}{(c^2+u)R(u)}. \quad (18)$$

Then, from equation (13), we get:

$$U_i=\pi abcG\varrho\ [D(0)-A(0)x_1^2-B(0)x_2^2-C(0)x_3^2]. \quad (19)$$

THE FAR-FIELD

At great distances ($\lambda > a$) from the ellipsoid, λ approaches r^2 and R(u) approaches $[u^3]^{0.5} = u^{1.5}$ for $u \geq \lambda$.

$$\therefore D(\lambda) \rightarrow \int_{r^2}^{\infty} du/u^{1.5} = [2/u^{0.5}]_{\infty}^{r^2} = 2/r.$$

Similarly, we have:

$$A(\lambda), B(\lambda), C(\lambda) \rightarrow \int_{r^2}^{\infty} du/u^{2.5} = 2/(3r^3)$$

Thus at large distances:

$$\begin{aligned} U_e \rightarrow \pi abcG\varrho[\,(2/r) - (2/3r^3)(x_1^2 + x_2^2 + x_3^2)\,] \\ = \pi abcG\varrho[\,(2/r) - (2/3r^3)r^2] \\ = (4\pi/3)abcG\varrho/r \end{aligned}$$

$$\therefore U_e \rightarrow G\varrho V/r = Gm_o/r, \text{ as r approaches } \infty \quad (20)$$

where $V = (4\pi/3)abc$ is the volume of the ellipsoid and $m_o = \varrho V$ is its mass. From equation (20) it is apparent that the gravitational potential, and hence also the field, asymptotically approaches that of a point particle of the same mass.

GRAVITATIONAL FIELD COMPONENTS

The external gravitational field components $\Delta g_j (j = 1,2,3)$ are obtained from equation (11) by differentiation:

$$\Delta g_1 = \frac{\partial U_e}{\partial x_1} = \pi abcG\varrho[[-(1 - \frac{x_1^2}{a^2+\lambda} - \frac{x_2^2}{b^2+\lambda} - \frac{x_3^2}{c^2+\lambda})/R(\lambda)]\ \frac{\partial\lambda}{\partial x_1} - 2x_1A(\lambda)].$$

By equation (2) the term in square brackets vanishes. Thus:

$$\Delta g_1 = -2\pi abcG\varrho x_1 A(\lambda). \quad (21)$$

Similarly, we get:

$$\Delta g_2 = -2\pi abcG\varrho x_2 B(\lambda) \quad (22)$$

$$\Delta g_3 = -2\pi abcG\varrho x_3 C(\lambda). \quad (23)$$

MAGNETIC SCALAR POTENTIAL AND FIELD COMPONENTS

The magnetic scalar potential Ω of a uniformly magnetized ellipsoid can be obtained from equations (14) and (19) via Poisson's relation (Grant & West 1965, p. 213), which may be written:

$$\Omega = -(1/G\varrho)\ \tilde{J}\ .\tilde{\nabla}U,$$

where $\tilde{J} = (J_1, J_2, J_3)$ is the magnetization vector.

$$\therefore \Omega_e = -(1/G\varrho)\tilde{J}\ .\Delta\tilde{g} = 2\pi abc[J_1x_1A(\lambda) + J_2x_2B(\lambda) + J_3x_3C(\lambda)\ ]. \quad (24)$$

Similarly, we get:

$$\Omega_i = 2\pi abc\ [J_1x_1A(0) + J_2x_2B(0) + J_3x_3C(0)\ ] \quad (25)$$

Equations (24) and (25) are consistent with formulae given by Stratton (1941) and Lowes (1974).

The external magnetic field components $\Delta B_j\ (= \Delta H_j)$ are obtained by differentiation ($\Delta\tilde{H} = -\tilde{\nabla}\,\Omega$). Thus, the external field components are:

$$\begin{aligned} \Delta B_1 = -2\pi abc\{J_1A(\lambda) + [J_1x_1A'(\lambda) + J_2x_2B'(\lambda) + \\ J_3x_3C'(\lambda)]\partial\lambda/\partial x_1\} \end{aligned} \quad (26)$$

$$\begin{aligned} \Delta B_2 = -2\pi abc\{J_2B(\lambda) + [J_1x_1A'(\lambda) + J_2x_2B'(\lambda) + \\ J_3x_3C'(\lambda)]\partial\lambda/\partial x_2\} \end{aligned} \quad (27)$$

$$\begin{aligned} \Delta B_3 = -2\pi abc\{J_3C(\lambda) + [J_1x_1A'(\lambda) + J_2x_2B'(\lambda) + \\ J_3x_3C'(\lambda)]\partial\lambda/\partial x_3\}, \end{aligned} \quad (28)$$

where, from equations (16), (17), (18), we have:

$$A'(\lambda) = -1/(a^2+\lambda)R(\lambda) \quad (29)$$

$$B'(\lambda) = -1/(b^2+\lambda)R(\lambda) \quad (30)$$

$$C'(\lambda) = -1/(c^2+\lambda)R(\lambda). \quad (31)$$

If a, b, c are distinct and finite, the expressions (15)–(18) involve incomplete elliptic integrals of the first and second kinds (Byrd & Friedman 1971, pp. 4–5). If two or more axes are equal, or if a approaches ∞, the integrals can be expressed in terms of elementary functions. Furthermore, in these cases the cubic equation for λ simplifies to a quadratic or linear equation.

THE INTERNAL MAGNETIC FIELD

The internal magnetic field arising from the magnetization is called the self-demagnetizing field and, from equation (25), has components:

$$\Delta H_1 = -\partial\Omega_i/\partial x_1 = -2\pi abc\ A(0)\ J_1 \quad (32)$$

$$\Delta H_2 = -\partial\Omega_i/\partial x_2 = -2\pi abc\ B(0)\ J_2 \quad (33)$$

$$\Delta H_3 = -\partial\Omega_i/\partial x_3 = -2\pi abc\ C(0)\ J_3. \quad (34)$$

So the internal field is independent of x_1, x_2, x_3; that is, it is uniform. It follows that the induced magnetization of a permeable (isotropic or anisotropic) ellipsoid in a uniform applied field is homogeneous because the resultant internal field (applied field plus self-demagnetizing field) is uniform. This is the reason for the importance of the ellipsoid model, particularly since the converse is also true for finite homogeneous bodies; only ellipsoids are truly uniformly magnetized by a uniform field.

Note that, for a triaxial ellipsoid, the self-demagnetizing field is not antiparallel to the magnetization unless $\tilde{J}$ is along one of the ellipsoid axes.

By comparison with the definition of the demagnetizing factors N_j', viz. $\Delta H_j = -N_j'J_j\ (j = 1, 2, 3)$, it can be seen from equations (32)–(34) that:

$$N_1' = 2\pi abc\ A(0) \quad (35)$$

$$N_2' = 2\pi abc\ B(0) \quad (36)$$

$$N_3' = 2\pi abc\ C(0). \quad (37)$$

CALCULATION OF RESULTANT MAGNETIZATION

The induced magnetization $\tilde{J}_{IND}$ of an ellipsoid in the ambient field $\tilde{F}$ is given by:

$$\tilde{J}_{IND} = K(\tilde{F} + \Delta\tilde{H}) = K(\tilde{F} - N\tilde{J}_R') = K(\tilde{F} - N\tilde{J}_{IND} - N\tilde{J}_{NRM}),$$

where K is the symmetric matrix of elements of the susceptibility tensor, $\tilde{J}_{NRM}$ is the remanence and $\tilde{J}_R'$ is the resultant (induced plus remanent) magnetization. Hence:

$$[I + KN]\tilde{J}_{IND} = K\tilde{F} - KN\tilde{J}_{NRM}$$

Whence $\tilde{J}_{IND} = [I + KN]^{-1}\ (K\tilde{F} - KN\tilde{J}_{NRM})$

$$\therefore \tilde{J}_R' = \tilde{J}_{IND} + \tilde{J}_{NRM} = [I + KN]^{-1}(K\tilde{F} - KN\tilde{J}_{NRM} + [I + KN]\tilde{J}_{NRM})$$

Thus:

$$\tilde{J}_R' = [I + KN]^{-1}\ (K\tilde{F} + \tilde{J}_{NRM}). \quad (38)$$

Magmod XV—the triaxial ellipsoid

The relevant notation is given in Table 1.

Table 1 Triaxial ellipsoid notation (refer to Fig. 1 and to the formulae, and to variables defined for previous MAGMODs in Emerson *et al* 1985)

C	centre of ellipsoid
Q	intersection of extrapolated major (a) axis of ellipsoid with x, y plane
Q′	intersection of extrapolated intermediate (b) axis of ellipsoid with x, y plane
a,b,c,	major, intermediate, minor semi-axes of ellipsoid (a>b>c)
α	azimuth of plunge of major axis (angle measured positive clockwise from + x axis to horizontal projection of downward-directed major axis), ($0° \le \alpha < 360°$)
δ	plunge of major axis, i.e. the angle between the major axis and its horizontal projection, ($0° \le \delta \le 90°$)
γ	angle between upward-directed intermediate axis and vertical plane containing major axis, positive clockwise looking along $\hat{v}_1$, ($-90° \le \gamma \le 90°$)
$\hat{v}_1$, $\hat{v}_2$, $\hat{v}_3$	unit vectors defining RH body axis co-ordinate system. $\hat{v}_1$ is directed along major axis in upward sense (i.e. parallel to $\tilde{CQ}$), with azimuth $\alpha - 180°$, inclination $-\delta$. $\hat{v}_2$ is directed along intermediate axis in upward sense (i.e. parallel to $\tilde{CQ}'$). $\hat{v}_3$ is directed along minor axis, such that $\hat{v}_3 = \hat{v}_1 \times \hat{v}_2$
l_i, m_i, n_i	direction cosines of $\hat{v}_i$ ($i = 1, 2, 3$) with respect to x, y, z axes
x_1, x_2, x_3	co-ordinates of observation point P with respect to body axes
$\tilde{r}$	directed distance from ellipsoid centre to observation point P($\tilde{r} = \tilde{CP}$, $r^2 = x^2 + h^2 = x_1^2 + x_2^2 + x_3^2$)
λ	an ellipsoidal co-ordinate, the largest root of the equation $x_1^2/(a^2+\lambda) + x_2^2/(b^2+\lambda) + x_3^2/(c^2+\lambda) = 1$
N_1', N_2', N_3'	demagnetizing factors along major, intermediate, minor axes

BODY AXES

$$\hat{v}_1 = (l_1, m_1, n_1) = (-\cos\alpha\cos\delta,\ -\sin\alpha\cos\delta,\ -\sin\delta)$$

$$\hat{v}_2 = (l_2, m_2, n_2) = (\cos\alpha\cos\gamma\sin\delta + \sin\alpha\sin\gamma,\ \sin\alpha\cos\gamma\sin\delta - \cos\alpha\sin\gamma, -\cos\gamma\cos\delta)$$

$$\hat{v}_3 = (l_3, m_3, n_3) = (\sin\alpha\cos\gamma - \cos\alpha\sin\gamma\sin\delta, -\cos\alpha\cos\gamma - \sin\alpha\sin\gamma\sin\delta,\ \sin\gamma\cos\delta)$$

The orientation of the ellipsoid may also be defined using the nomenclature of structural geology. The attitude of the ellipsoid is determined unambiguously if the strike and dip of the a, b plane are specified together with the rake, or pitch, of the a-axis within this plane, measured from the strike direction. To simplify the mathematical relationships between the alternative methods of defining the orientation the convention that the strike of the a, b plane has right-to-left sense when viewed with the plane dipping towards the observer is adopted (refer to Fig. 1). Thus the dip is always ≤ 90°. The unit vectors of the body axis co-ordinate system are then given by:

$$\hat{v}_1 = (-\cos STRIKE\cos RAKE - \sin STRIKE\cos DIP\sin RAKE, -\sin STRIKE\cos RAKE + \cos STRIKE\cos DIP\sin RAKE,\ -\sin DIP\sin RAKE)$$

$$\hat{v}_2 = \pm(\cos STRIKE\sin RAKE - \sin STRIKE\cos DIP\cos RAKE,\ \sin STRIKE\sin RAKE + \cos STRIKE\cos DIP\cos RAKE,\ -\sin DIP\cos RAKE)$$

$$\hat{v}_3 = \pm\ (\sin STRIKE\sin DIP,\ -\cos STRIKE\sin DIP,\ -\sin DIP),$$

where the plus signs apply for $0° \le RAKE \le 90°$ and the minus signs if $90° < RAKE < 180°$.

In terms of the parameters α, δ, γ the structural angles are given by:

$$RAKE = \cos^{-1}[-sgn(\gamma)\cos\delta/\surd(1 + \tan^2\gamma\sin^2\delta)],\ 0° \le RAKE < 180°$$

$$STRIKE = \alpha + \cos^{-1}[-sgn(\gamma)/\surd(1 + \tan^2\gamma\sin^2\delta)]$$

$$DIP = \sin^{-1}[\cos\gamma\surd(1 + \tan^2\gamma\sin^2\delta)],\ 0° \le DIP \le 90°,$$

$$\text{where } sgn(\gamma) = \begin{cases} +1, & \gamma > 0 \\ 0, & \gamma = 0 \\ -1, & \gamma < 0. \end{cases}$$

The inverse relationships are given in Fig. 1.

CO-ORDINATES WITH RESPECT TO BODY AXES

Along a principal profile the body axis co-ordinates are:

$$x_1 = xl_1 - hn_1$$

$$x_2 = xl_2 - hn_2$$

$$x_3 = xl_3 - hn_3.$$



Fig. 1 MAGMOD XV triaxial ellipsoid.

At a point $(x, y, 0)$ in the horizontal plane passing through the origin:

$$x_j = x l_j + y m_j - h n_j, \qquad (j = 1, 2, 3).$$

Over an irregular ground surface the body axis co-ordinates are given by

$$x_j = x l_j + y m_j - (h + \Delta h) n_j, \qquad (j = 1, 2, 3)$$

where Δh is the elevation of the observation point above the origin, which lies on the irregular surface at height h directly above the centre of the ellipsoid.

If the geographic co-ordinates of an observation point in a drillhole are (x, y, h') the body axis co-ordinates are:

$$x_j = x l_j + y m_j + (h' - h) n_j, \qquad (j = 1, 2, 3).$$

Here, h' is the depth of the observation point below the origin.

ELLIPSOIDAL CO-ORDINATE λ AND ITS SPATIAL DERIVATIVES

The co-ordinate λ is the largest root of the cubic equation:

$$\lambda^3 + p_2\lambda^2 + p_1\lambda + p_0 = 0$$

and is given by equations (6) and (7). The spatial derivatives of λ are given by equation (10) etc.

DEMAGNETIZING FACTORS

$$N'_1 = \frac{4\pi abc}{(a^2 - b^2)(a^2 - c^2)^{0.5}}[F(k,\theta) - E(k,\theta)]$$



Fig. 2 Worked example: MAGMOD XV including demagnetisation, susceptibility, anisotropy and remanence.

$$N_2' = \frac{4\pi abc(a^2-c^2)^{0.5}}{(a^2-b^2)(b^2-c^2)}\left[E(k,\theta) - \left(\frac{b^2-c^2}{a^2-c^2}\right)F(k,\theta) - \frac{c(a^2-b^2)}{ab(a^2-c^2)^{0.5}}\right]$$

$$N_3' = \frac{4\pi abc}{(b^2-c^2)(a^2-c^2)^{0.5}}\left[\frac{b(a^2-c^2)^{0.5}}{ac} - E(k,\theta)\right]$$

where

$$k = \left(\frac{a^2-b^2}{a^2-c^2}\right)^{0.5}, \cos\theta = c/a \; (0 \leq \theta \leq \pi/2).$$

SUSCEPTIBILITY TENSOR AND MATRIX

$$k_{ij} = \sum_r k_r(L_r l_i + M_r m_i + N_r n_i)(L_r l_j + M_r m_j + N_r n_j) \qquad (r = 1, 2, 3)$$

$$K = [k_{ij}]$$

FIELD AND NRM COMPONENTS WITH RESPECT TO BODY AXES

$$\tilde{F} = \begin{bmatrix} F_1 \\ F_2 \\ F_3 \end{bmatrix}, \quad \tilde{J}_{NRM} = \begin{bmatrix} (J_N)_1 \\ (J_N)_2 \\ (J_N)_3 \end{bmatrix}$$

$$F_i = F(l l_i + m m_i + n n_i)$$

$$(J_N)_i = J_N(l_N l_i + m_N m_i + n_N n_i)$$

RESULTANT MAGNETIZATION WITH RESPECT TO BODY AXES

If self-demagnetization is neglected, the resultant magnetization is given by:

$$\tilde{J}_R = K\tilde{F} + \tilde{J}_{NRM.}$$

Let

$$A = I + KN = \begin{bmatrix} 1+k_{11}N_1' & k_{12}N_2' & k_{13}N_3' \\ k_{12}N_1' & 1+k_{22}N_2' & k_{23}N_3' \\ k_{13}N_1' & k_{23}N_2' & 1+k_{33}N_3' \end{bmatrix}$$

The resultant magnetization corrected for self-demagnetization is then given by:

$$\tilde{J}_R' = A^{-1}\tilde{J}_R.$$

ANOMALOUS FIELD COMPONENTS WITH RESPECT TO BODY AXES

Provided a > b > c, the components of the external field due to the ellipsoid are:

$$\Delta B_1 = f_1 \frac{\partial\lambda}{\partial x_1} - 2\pi abc\, J_1'\, A(\lambda)$$

$$\Delta B_2 = f_1 \frac{\partial\lambda}{\partial x_2} - 2\pi abc\, J_2'\, B(\lambda)$$

$$\Delta B_3 = f_1 \frac{\partial\lambda}{\partial x_3} - 2\pi abc\, J_3'\, C(\lambda)$$

where

$$f_1 = \frac{2\pi abc}{[(a^2+\lambda)(b^2+\lambda)(c^2+\lambda)]^{0.5}} \left[\frac{J_1'x_1}{a^2+\lambda} + \frac{J_2'x_2}{b^2+\lambda} + \frac{J_3'x_3}{c^2+\lambda}\right]$$

$$A(\lambda) = \frac{2}{(a^2-b^2)(a^2-c^2)^{0.5}} \left[F(k,\theta') - E(k,\theta')\right]$$

$$B(\lambda) = \frac{2(a^2-c^2)^{0.5}}{(a^2-b^2)(b^2-c^2)} \left[E(k,\theta') - \left(\frac{b^2-c^2}{a^2-c^2}\right) F(k,\theta') - \frac{k^2\sin\theta'\cos\theta'}{(1-k^2\sin^2\theta')^{0.5}} \right]$$

$$C(\lambda) = \frac{2}{(b^2-c^2)(a^2-c^2)^{0.5}}\left[\frac{\sin\theta'(1-k^2\sin^2\theta')^{0.5}}{\cos\theta'} - E(k,\theta')\right]$$

$$\sin\theta' = \left(\frac{a^2-c^2}{a^2+\lambda}\right)^{0.5} \quad (0 \le \theta' \le \pi/2).$$

$F(k,\theta')$ and $E(k,\theta')$ are, respectively, Legendre's normal elliptic integrals of the first and second kind (Byrd & Friedman 1971).

Several methods for calculation of the integrals (16)–(18) have been published; for example, papers by Carlson (1979) and Carlson and Notis (1981). The HP41CX program listed at the end of this paper uses an algorithm due to Bulirsch (1965) for calculation of A(λ), B(λ) and C(λ).

ANOMALOUS FIELD COMPONENTS WITH RESPECT TO GEOGRAPHIC AXES

With respect to geographic (*x*, *y*, *z*) axes the anomalous field components are:

$$\Delta B_x = \Delta B_1 l_1 + \Delta B_2 l_2 + \Delta B_3 l_3$$

$$\Delta B_y = \Delta B_1 m_1 + \Delta B_2 m_2 + \Delta B_3 m_3$$

$$\Delta B_z = \Delta B_1 n_1 + \Delta B_2 n_2 + \Delta B_3 n_3.$$

The component of the anomalous field projected onto the regional magnetic meridian is:

$$\Delta B_H = \Delta B_x \cos\beta + \Delta B_y \sin\beta.$$

The component of the anomalous field vector projected onto the regional geomagnetic field $\bar{F}$ is:

$$\Delta B_T = \Delta B_H \cos I + \Delta B_z \sin I.$$

The total field anomaly measured by a sensor that responds only to the magnitude of the field is:

$$\Delta B_m = [(F_x+\Delta B_x)^2 + (F_y+\Delta B_y)^2 + (F_z+\Delta B_z)^2]^{0.5} - F.$$

GRAVMOD XV – triaxial ellipsoid

The diagrams and symbols for GRAVMODS XV, XIA, XIB and XII are as for the corresponding MAGMODS (Emerson *et al* 1985) with the addition of ϱ (density contrast) and G (gravitational constant). The anomalous gravity components with respect to body axes are given by equations (21)–(23). The observed gravity effect is:

$$\Delta g_z = \Delta g_1 n_1 + \Delta g_2 n_2 + \Delta g_3 n_3.$$

GRAVMOD XIA – Prolate ellipsoid of revolution

If b = c < a, then the gravity effect components with respect to body axes are given by:

$$\Delta g_1 = \frac{4\pi ab^2 G\varrho}{(a^2-b^2)^{1.5}} x_1 \left\{\left[\frac{a^2-b^2}{a^2+\lambda}\right]^{0.5} - \log_e\left[\frac{(a^2-b^2)^{0.5}+(a^2+\lambda)^{0.5}}{(b^2+\lambda)^{0.5}}\right]\right\}$$

$$\Delta g_2 = \frac{2\pi ab^2 G\varrho}{(a^2-b^2)^{1.5}} x_2 \left\{\log_e\left[\frac{(a^2-b^2)^{0.5}+(a^2+\lambda)^{0.5}}{(b^2+\lambda)^{0.5}}\right] - \frac{[(a^2-b^2)(a^2+\lambda)]^{0.5}}{(b^2+\lambda)}\right\} = G\varrho x_2 f_2$$

$$\Delta g_3 = G\varrho x_3 f_2.$$

The co-ordinate λ is the same for GRAVMOD XIA, GRAVMOD XIB and MAGMOD XIA (see Emerson *et al* 1985, p. 49).

The observed gravity effect is, as above:

$$\Delta g_z = \Delta g_1 n_1 + \Delta g_2 n_2 + \Delta g_3 n_3.$$

GRAVMOD XIB – Oblate ellipsoid of revolution

The gravity effect with respect to body axes is:

$$\Delta g_1 = \frac{4\pi ab^2 G\varrho}{(b_2-a^2)^{1.5}} x_1 \left\{\tan^{-1}\left[\frac{b^2-a^2}{a^2+\lambda}\right]^{0.5} - \left[\frac{b^2-a^2}{a^2+\lambda}\right]^{0.5}\right\}$$

$$\Delta g_2 = \frac{2\pi ab^2 G\varrho}{(b^2-a^2)^{1.5}} x_2 \left\{\frac{[(b^2-a^2)(a^2+\lambda)]^{0.5}}{b^2+\lambda} - \tan^{-1}\left[\frac{b^2-a^2}{a^2+\lambda}\right]^{0.5}\right\} = G\varrho x_2 f_2$$

$$\Delta g_3 = \frac{2\pi ab^2 G\varrho}{(b^2-a^2)^{1.5}} x_3 \left\{\frac{[(b^2-a^2)(a^2+\lambda)]^{0.5}}{b^2+\lambda} - \tan^{-1}\left[\frac{b^2-a^2}{a^2+\lambda}\right]^{0.5}\right\} = G\varrho x_3 f_2.$$

The observed gravity effect is:

$$\Delta g_z = \Delta g_1 n_1 + \Delta g_2 n_2 + \Delta g_3 n_3.$$

GRAVMOD XII – 2D elliptic cylinder

The gravity effect components with respect to body axes are:

$$\Delta g_1 = 0$$

$$\Delta g_2 = \frac{4\pi bcG\varrho}{(b^2-c^2)} \quad x_2 \left[\left(\frac{c^2+\lambda}{b^2+\lambda} \right)^{0.5} - 1 \right]$$

$$\Delta g_3 = \frac{4\pi bcG\varrho}{(b^2-c^2)} \quad x_3 \left[1 - \left(\frac{b^2+\lambda}{c^2+\lambda} \right)^{0.5} \right].$$

The co-ordinate λ is the same for GRAVMOD XII as for MAGMOD XII (see Emerson *et al* 1985, p. 60).

The observed gravity effect is:

$$\Delta g_z = \Delta g_2 n_2 + \Delta g_3 n_3.$$

Acknowledgment

We thank Maurice Craig for his careful scrutiny of the manuscript.

References

Bulirsch R. (1965), 'Numerical calculation of elliptic integrals and elliptic functions', *Numerische Mathematik* 7, 78–90.

Byrd P. F. & Friedman M. D. (1971), *Handbook of Elliptic Integrals for Engineers and Scientists*. Springer-Verlag, Berlin.

Carlson B. C. (1979), 'Computing elliptic integrals by duplication', *Numerische Mathematik* 33, 1–16.

Carlson B. C. & Notis E. M. (1981), 'Algorithms for incomplete elliptic integrals', *ACM Trans. Math. Soft.* 7, 398–403.

Dirichlet P.G. L. (1839), 'Sur une novelle méthode pour la détermination des intégrales multiples,' *Journal de Mathematiques (Liouville) Sér.* 4, 164–8

Emerson D. W., Clark D. A. & Saul S. J. (1985), 'Magnetic exploration models incorporating remanence, demagnetization and anisotropy: HP41C handheld computer algorithms', *Explor. Geophys.* 16, 1–122.

Farrar L. J. D. (1979), 'Some comments on detailed magnetic investigations of ellipsoidal bodies at Tennant Creek', *Bull. Aust. Soc. Explor. Geophys.* 10, 26–33.

Gaal G. (1977), 'Structural features of Precambrian, stratabound sulphide-ore deposits in Finland', *Geologiska Foreningens i Stockholm Forhandlingar* 99, 118–26.

Grant F. S. & West G. F. (1965), *Interpretation Theory in Applied Geophysics*. McGraw-Hill, New York.

Hjelt S. E. & Turunen P. (1981), 'Magnetostatic fields of some bodies of simple geometric form', *in*: S. E. Hjelt & A. Ph Phokin , eds, *Interpretation of Borehole Magnetic Data and Some Special Problems of Magnetometry*, Department of Geophysics, University of Oulu, Report no. 1, pp. 5–14.

Kellogg O. D. (1929), *Foundations of Potential Theory*. Frederick Ungar, New York.

Lowes F. J. (1974), 'The torque on a magnet', *Proc. Roy. Soc. Lond. A.* 337, 555–67.

Pedersen L. B. (1985), 'The gravity and magnetic fields from ellipsoidal bodies in the wavenumber domain', *Geophys. Prosp.* 33, 263–81.

Solomon M., Schmidt B. L. & Walshe J. L. (1986), 'The base metal deposits of the Cobar district, Australia – are they further examples of syntectonic, synmetamorphic ore deposits?' *BMR J. Geol. Geophys.* (in press).

Stratton, J. A. (1941), *Electromagnetic Theory*. McGraw-Hill, New York.

NOTES ON HP41C PROGRAMS FOR MAGMOD XV

1. *Users of these programs are on their own and should clearly understand that the program material contained herein is supplied without representation or warranty of any kind. Despite the extensive checking and editing that preceded publication, the authors, the ASEG, the publishers, and equipment manufacturers accept or assume no responsibility whatsoever and shall have no liability, consequential or otherwise, for the use, performance and results of these programs or for any actions that may follow the usage of these programs. No claims whatsoever are made or implied regarding the elegance or accuracy of the programs. Errors may come to light as the programs are used; users should document any such errors and notify the ASEG so that revisions can be made. The programs were designed as a learning aid for the teaching and understanding of applied magnetics. They are not intended to compete with sophisticated large computer packages. However, they may be useful to practising professional field geophysicists in preliminary analyses of magnetic data.*
2. Refer to: Emerson, Clark & Saul (1985) *Explor. Geophys.* **16**, 1–156, for general information on the MAGMOD suite; also see: Corrigenda and Addenda, *Explor. Geophys.* **16**, 395.
3. PROGRAMS
   ELI: elliptic integral calculations
   C1: sets up parameters for ELI during anomaly computation
   C2: takes elliptic integrals and calculates anomalies
   MAG15: control program
4. Procedure for loading program cards, follow a similar procedure if manually entering the programs.

```
XEQ α SIZE α
SIZE 000
LOAD PROG MAG15
(SHIFT KEY) GTO ..
LOAD PROG ELI
(SHIFT KEY) GTO ..
LOAD PROG C1
(SHIFT KEY) GTO ..
LOAD PROG C2
(SHIFT KEY) GTO ..
α ELI α
XEQ α SAVEP α
α C1 α
XEQ α SAVEP α
α C2 α
XEQ α SAVEP α
XEQ α CLP α
α C1 α
XEQ α CLP α
α C2 α
XEQ α CLP α
```

α ELI α
XEQ α SIZE α
SIZE 067
XEQ α MAG 15 α

5. DO NOT use (shift key) GTO .. once programs are in and running
6. Avoid extreme a:b:c ellipsoid semi axis values
7. In USER INSTRUCTIONS add: Profile recomputation with changed inputs similar to previously published MAGMODS except no plot; for MAGMOD XV to obtain resultant magnetization and β_R press G. (see: *Explor. Geophys.* **16**, 1985).
8. HP41C computer will need quad memory module and extended functions module.
HP41CV computer will need extended functions module.
HP41CX computer will handle as is.

User Instructions

MAGMOD XV: TRIAXIAL ELLIPSOID: DEMAGNETISATION, ANISOTROPIC SUSCEPTIBILITY, REMANENCE (card sides 1, 2)

User mode
Deg mode
Size 067
(see notes)

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Load cards, run program		XEQ	MAG15	F =
2	Geomagnetic field magnitude	F		R/S	I =
3	Geomagnetic field inclination	I		R/S	D =
4	Geomagnetic field declination	D		R/S	DEPTH =
5	Depth to ellipsoid centre	h		R/S	PLUNGE =
6	Plunge of major axis	δ		R/S	AZIMUTH =
7	Azimuth of plunge axis	α		R/S	TILT =
8	Tilt of intermediate axis	γ		R/S	a =
9	Major semi axis of ellipsoid	a		R/S	b =
10	Intermediate semi axis of ellipsoid	b		R/S	c =
11	Minor semi axis of ellipsoid	c		R/S	
	Observe ellipsoid volume on printout				Ka =
12	Input magnitude of maximum susceptibility	k_a		R/S	D =
13	Input declination of maximum susceptibility	D_a		R/S	I =
14	Input inclination of maximum susceptibility	I_a		R/S	Kb =
15	Input magnitude of intermediate susceptibility	k_b		R/S	D =
16	Input declination of intermediate susceptibility	D_d		R/S	I =
17	Input inclination of intermediate susceptibility	I_b		R/S	Kc =
18	Input magnitude of minimum susceptibility	kc		R/S	D =
19	Input declination of minimum susceptibility	Dc		R/S	I =
20	Input inclination of minimum susceptibility	Ic		R/S	REM ?<0,1>
	Notes (i)susceptibility axes must be orthogonal				
	(ii)if isotropic susceptibility, insert				
	$k_a=k_b=k_c$ with orthogonal axes e.g. D,I:0,90;0,0;90,0				
21	If remanence absent	0		R/S	BEARING =
22	If remanence present	1		R/S	JREM =
23	Remanent magnetisation magnitude	J_{REM}		R/S	IREM =
24	Remanent magnetisation inclination	I_{REM}		R/S	DREM =
25	Remanent magnetisation declination	D_{REM}		R/S	BEARING =
26	Azimuth of magnetic north w.r.t. + x axis	β		R/S	
27	Observe demagnetisation factors Na, Nb, Nc				N1 =
					XMIN = ?
28	Minimum (profile) x value	X_{MIN}		R/S	XMAX = ?
29	Maximum (profile) x value	X_{MAX}		R/S	XINC = ?
30	Profile x increment	X_{INC}		R/S	
31	Observe printout of: station				X =
32	vertical component anomaly				BZ =
33	total intensity anomaly				BT =
34	Resultant magnetisation calculation			G	J,I,D,β RES

HP41C
TRIAXIAL
ELLIPSOID
PROGRAM
MAGNETICS
MAGMOD XV

```
01*LBL "MAG15"
02*LBL A
03 CF 22
04 FIX 1
05 "F="
06 PROMPT
07 FS?C 22
08 STO 03
09 ARCL 03
10 PRA
11 "I="
12 PROMPT
13 FS?C 22
14 STO 04
15 ARCL 04
16 PRA
17 "D="
18 PROMPT
19 FS?C 22
20 STO 05
21 ARCL 05
22 PRA
23 ADV
24*LBL E
25 "DEPTH="
26 PROMPT
27 FS?C 22
28 STO 13
29 ARCL 13
30 PRA
31 "PLUNGE="
32 PROMPT
33 FS?C 22
34 STO 14
35 ARCL 14
36 PRA
37 "AZIMUTH="
38 PROMPT
39 FS?C 22
40 STO 15
41 ARCL 15
42 PRA
43 "TILT="
44 PROMPT
45 FS?C 22
46 STO 16
47 ARCL 16
48 PRA
49 "a="
50 PROMPT
51 FS?C 22
52 STO 17
53 ARCL 17
54 PRA
55 "b="
56 PROMPT
57 FS?C 22
58 STO 18
59 ARCL 18
60 PRA
61 "c="
62 PROMPT
63 FS?C 22
64 STO 19
65 ARCL 19
66 PRA
67 RCL 17
68 ENTER↑
69 X↑2
70 RCL 18
71 X↑2
72 -
73 STO 31
74 STO 35
75 RDN
76 RCL 18
77 *
78 RCL 19
79 *
80 4
81 *
82 PI
83 *
84 STO 32
85 3
86 /
87 "VOL="
88 ARCL X
89 PRA
90 ADV
91 RCL 18
92 X↑2
93 RCL 19
94 X↑2
95 -
96 STO 33
97 RCL 31
98 +
99 STO 34
100 ST/ 35
101 RCL 15
102 COS
103 STO 00
104 STO 36
105 STO 37
106 STO 41
107 LASTX
108 SIN
109 STO 01
110 STO 38
111 STO 39
112 STO 40
113 RCL 14
114 COS
115 STO 44
116 CHS
117 ST* 36
118 ST* 39
119 STO 43
120 LASTX
121 SIN
122 ST* 37
123 ST* 40
124 CHS
125 STO 42
126 RCL 37
127 RCL 40
128 RCL 16
129 COS
130 ST* 37
131 ST* 38
132 ST* 40
133 ST* 43
134 CHS
135 ST* 41
136 CLX
137 LASTX
138 SIN
139 ST* 00
140 ST* 01
141 ST* 44
142 *
143 ST- 41
144 CLX
145 LASTX
146 *
147 ST- 38
148 RCL 00
149 ST- 40
150 RCL 01
151 ST* 37
152 FIX 6
153 "Ka="
154 PROMPT
155 FS?C 22
156 STO 20
157 ARCL 20
158 PRA
159 FIX 1
160 "D="
161 PROMPT
162 FS?C 22
163 STO 21
164 ARCL 21
165 PRA
166 "I="
167 PROMPT
168 FS?C 22
169 STO 22
170 ARCL 22
171 PRA
172 ADV
173 FIX 6
174 "Kb="
175 PROMPT
176 FS?C 22
177 STO 23
178 ARCL 23
179 PRA
180 FIX 1
181 "D="
182 PROMPT
183 FS?C 22
184 STO 24
185 ARCL 24
186 PRA
187 "I="
188 PROMPT
189 FS?C 22
190 STO 25
191 ARCL 25
192 PRA
193 ADV
194 FIX 6
195 "Kc="
196 PROMPT
197 FS?C 22
198 STO 26
199 ARCL 26
200 PRA
201 FIX 1
202 "D="
203 PROMPT
204 FS?C 22
205 STO 27
206 ARCL 27
207 PRA
208 "I="
209 PROMPT
210 FS?C 22
211 STO 28
212 ARCL 28
213 PRA
214 ADV
215 "REM?(0,1)"
216 PROMPT
217 CF 22
218 CF 05
219 X=0?
220 GTO D
221*LBL C
222 SF 05
223 "J REM="
224 PROMPT
225 FS?C 22
226 STO 09
227 ARCL 09
228 PRA
229 "I REM="
230 PROMPT
231 FS?C 22
232 STO 07
233 ARCL 07
234 PRA
235 "D REM="
236 PROMPT
237 FS?C 22
238 STO 08
239 ARCL 08
240 PRA
241 ADV
242*LBL B
243 "BEARING="
244 PROMPT
245 FS?C 22
246 STO 29
247 ARCL 29
248 PRA
249 ADV
250 1
251 RCL 22
252 RCL 21
253 XEQ 01
254 STO 00
255 X<>Y
256 STO 01
257 R↑
258 STO 02
259 1
260 RCL 25
261 RCL 24
262 XEQ 01
263 STO 06
264 X<>Y
265 STO 10
266 R↑
267 STO 11
268 1
269 RCL 28
270 RCL 27
271 XEQ 01
272 STO 12
273 X<>Y
274 STO 30
275 R↑
276 STO 49
277 RCL 42
278 *
279 RCL 30
280 RCL 39
281 *
282 +
283 RCL 12
284 RCL 36
285 *
286 +
287 STO 50
288 RCL 12
289 RCL 37
290 *
291 RCL 30
292 RCL 40
293 *
294 +
295 RCL 49
296 RCL 43
297 *
298 +
299 STO 51
300 RCL 38
301 ST* 12
302 RCL 30
303 RCL 41
304 *
305 RCL 49
306 RCL 44
307 *
308 +
309 ST+ 12
310 RCL 00
311 RCL 36
312 *
313 RCL 01
314 RCL 39
315 *
316 +
317 RCL 02
318 RCL 42
319 *
320 +
321 STO 30
322 RCL 00
323 RCL 37
324 *
325 RCL 01
326 RCL 40
327 *
328 +
329 RCL 02
330 RCL 43
331 *
332 +
333 STO 49
334 RCL 38
335 ST* 00
336 RCL 01
337 RCL 41
338 *
339 RCL 02
340 RCL 44
341 *
342 +
343 ST+ 00
344 RCL 06
345 RCL 36
346 *
347 RCL 10
348 RCL 39
349 *
350 +
351 RCL 11
352 RCL 42
353 *
354 +
355 STO 01
356 RCL 06
357 RCL 37
358 *
359 RCL 10
360 RCL 40
361 *
362 +
363 RCL 11
364 RCL 43
365 *
366 +
367 STO 02
368 RCL 38
369 ST* 06
370 RCL 10
371 RCL 41
372 *
373 RCL 11
374 RCL 44
375 *
376 +
377 ST+ 06
378 RCL 20
379 RCL 30
380 X↑2
381 *
382 RCL 23
383 RCL 01
384 X↑2
385 *
386 +
387 RCL 26
388 RCL 50
389 X↑2
390 *
391 +
392 STO 52
393 RCL 20
394 RCL 30
395 *
396 RCL 49
397 *
398 RCL 23
399 RCL 01
400 *
401 RCL 02
402 *
403 +
404 RCL 26
405 RCL 50
406 *
407 RCL 51
408 *
409 +
410 STO 53
411 RCL 20
412 RCL 00
413 *
414 ST* 30
415 RCL 23
416 RCL 01
417 *
418 RCL 06
419 *
420 RCL 26
421 RCL 50
422 *
423 RCL 12
424 *
425 +
426 ST+ 30
427 RCL 20
428 RCL 49
429 X↑2
430 *
431 RCL 23
432 RCL 02
433 X↑2
434 *
435 +
436 RCL 26
437 RCL 51
438 X↑2
439 *
440 +
441 STO 01
442 RCL 20
443 RCL 00
444 *
445 ST* 49
446 RCL 23
447 RCL 02
448 *
449 RCL 06
450 *
451 RCL 26
452 RCL 51
453 *
454 RCL 12
455 *
456 +
457 ST+ 49
458 RCL 20
459 RCL 00
460 *
461 ST* 00
462 RCL 23
463 RCL 06
464 X↑2
465 *
466 RCL 26
467 RCL 12
468 X↑2
469 *
470 +
471 ST+ 00
472 RCL 29
473 RCL 04
474 RCL 03
475 XEQ 02
476 STO 10
477 X<>Y
478 STO 11
479 R↑
480 STO 12
481 RCL 42
482 *
483 X<>Y
484 RCL 39
485 *
486 +
487 X<>Y
488 RCL 36
```

```
489 *
490 +
491 STO 06
492 RCL 43
493 RCL 12
494 *
495 RCL 10
496 RCL 37
497 *
498 +
499 RCL 11
500 RCL 40
501 *
502 +
503 X<> 12
504 RCL 41
505 ST* 11
506 X<>Y
507 RCL 44
508 *
509 RCL 10
510 RCL 38
511 *
512 +
513 ST+ 11
514 RCL 06
515 RCL 52
516 *
517 RCL 12
518 RCL 53
519 *
520 +
521 RCL 11
522 RCL 30
523 *
524 +
525 STO 10
526 RCL 06
527 RCL 53
528 *
529 RCL 12
530 RCL 01
531 *
532 +
533 RCL 11
534 RCL 49
535 *
536 +
537 X<> 11
538 RCL 00
539 *
540 RCL 06
541 RCL 30
542 *
543 +
544 RCL 12
545 RCL 49
546 *
547 +
548 STO 12
549 FC? 05
550 GTO 03
551 RCL 09
552 RCL 07
553 RCL 08
554 SF 06
555*LBL 01
556 RCL 29
557 +
558 RCL 05
559 -
560 X<> Z
561*LBL 02
562 P-R
563 X<>Y
564 RDN
565 P-R
566 FC?C 06
567 RTN
568 STO 06
569 X<>Y
570 STO 02
571 R↑
572 STO 50
573 RCL 42
574 *
575 X<>Y
576 RCL 39
577 *
578 +
579 X<>Y
580 RCL 36
581 *
582 +
583 ST+ 10
584 RCL 06
585 RCL 37
586 *
587 RCL 02
588 RCL 40
589 *
590 +
591 RCL 50
592 RCL 43
593 *
594 +
595 ST+ 11
596 RCL 06
597 RCL 38
598 *
599 RCL 02
600 RCL 41
601 *
602 +
603 RCL 50
604 RCL 44
605 *
606 +
607 ST+ 12
608*LBL 03
609 RCL 34
610 RCL 17
611 X↑2
612 /
613 STO 65
614 SQRT
615 ASIN
616 TAN
617 STO 06
618 STO 60
619 "ELI"
620 GETP
621 XEQ "ELI"
622 RCL 32
623 ST* 06
624 ST* 51
625 CHS
626 ST* 50
627 RCL 51
628 "H1="
629 FIX 4
630 ARCL X
631 PRA
632 ST* 52
633 RCL 30
634 *
635 X<> 30
636 RCL 49
637 RCL 53
638 ST* 51
639 RDN
640 RCL 50
641 "H2="
642 ARCL Y
643 PRA
644 ST* 01
645 ST* 53
646 *
647 STO 50
648 RDN
649 RCL 06
650 "H3="
651 ARCL X
652 PRA
653 ADV
654 FIX 1
655 ST* 00
656 ST* 49
657 *
658 STO 06
659 1
660 ST+ 52
661 ST+ 01
662 ST+ 00
663 RCL 01
664 RCL 00
665 *
666 RCL 49
667 RCL 50
668 *
669 -
670 STO 54
671 RCL 52
672 *
673 RCL 49
674 RCL 30
675 *
676 RCL 51
677 RCL 00
678 *
679 -
680 STO 55
681 RCL 53
682 *
683 +
684 RCL 51
685 RCL 50
686 *
687 RCL 01
688 RCL 30
689 *
690 -
691 STO 56
692 RCL 06
693 *
694 +
695 STO 59
696 RCL 53
697 RCL 49
698 *
699 RCL 06
700 RCL 01
701 *
702 -
703 STO 57
704 RCL 52
705 ST* 01
706 RCL 53
707 RCL 51
708 *
709 ST- 01
710 RCL 06
711 ST* 51
712 RCL 52
713 RCL 49
714 *
715 ST- 51
716 RCL 52
717 RCL 00
718 *
719 RCL 06
720 RCL 30
721 *
722 -
723 STO 58
724 RCL 50
725 ST* 06
726 RCL 53
727 RCL 00
728 *
729 ST- 06
730 RCL 53
731 ST* 30
732 RCL 52
733 RCL 50
734 *
735 ST- 30
736 RCL 54
737 RCL 10
738 *
739 RCL 06
740 RCL 11
741 *
742 +
743 RCL 57
744 RCL 12
745 *
746 +
747 RCL 55
748 RCL 10
749 *
750 RCL 58
751 RCL 11
752 *
753 +
754 RCL 51
755 RCL 12
756 *
757 +
758 X<> 11
759 RCL 39
760 *
761 RCL 56
762 RCL 10
763 *
764 +
765 RCL 01
766 RCL 12
767 *
768 +
769 STO 12
770 X<>Y
771 RCL 59
772 ST/ 11
773 ST/ 12
774 /
775 STO 10
776*LBL E
777 "XMIN=?"
778 PROMPT
779 FS?C 22
780 STO 45
781 "XMAX=?"
782 PROMPT
783 FS?C 22
784 STO 47
785 "XINC=?"
786 PROMPT
787 FS?C 22
788 STO 48
789 RCL 45
790 STO 46
791*LBL 04
792 "X="
793 ARCL 46
794 PRA
795 XEQ 01
796 "BZ="
797 ARCL 52
798 PRA
799 "BT="
800 ARCL X
801 PRA
802 ADV
803 RCL 48
804 ST+ 46
805 RCL 47
806 RCL 46
807 X<=Y?
808 GTO 04
809 STOP
810 GTO E
811*LBL 01
812 "C1"
813 GETP
814 XEQ "C1"
815 "ELI"
816 GETP
817 XEQ "ELI"
818 "C2"
819 GETP
820 XEQ "C2"
821 RTN
822*LBL G
823 "C2"
824 GETP
825 XEQ "GG"
826 END
```

HP41C
TRIAXIAL
ELLIPSOID
PROGRAM
MAGNETICS
ELI

```
01*LBL "ELI"
02 1
03 STO 50
04 STO 51
05 STO 54
06 STO 55
07 STO 56
08 RCL 35
09 -
10 STO 57
11 STO 58
12 SQRT
13 STO 59
14 RCL 60
15 ST/ 50
16 X↑2
17 ST* 58
18 1
19 ST+ 58
20 +
21 ST/ 58
22 ST/ 60
23 0
24 STO 61
25 STO 62
26 STO 63
27 RCL 58
28 SQRT
29 STO 58
30*LBL 06
31 RCL 55
32 RCL 56
33 RCL 51
34 /
35 ST+ 55
36 RDN
37 RCL 59
38 *
39 ST+ 56
40 RCL 54
41 RCL 57
42 RCL 51
43 /
44 ST+ 54
45 RDN
46 RCL 59
47 *
48 ST+ 57
49 RCL 51
50 RCL 59
51 *
52 STO 64
53 RCL 50
54 /
55 ST- 50
56 RCL 60
57 RCL 58
58 /
59 RCL 61
60 RCL 64
61 RCL 58
62 /
63 *
64 ST+ 60
65 LASTX
66 ST+ 58
67 X<> Z
68 ST+ 61
69 2
70 ST/ 55
71 ST/ 54
72 ST/ 61
73 1
74 ST+ 63
75 RCL 50
76 SIGN
77 -
78 2
79 RCL 63
80 Y↑X
81 /
82 ST+ 62
83 1
84 RCL 59
85 RCL 51
86 RCL 59
87 +
88 X<> 51
89 /
90 -
91 1 E-6
92 X>Y?
93 GTO 01
94 RCL 64
95 SQRT
96 2
97 *
98 STO 59
99 GTO 06
100*LBL 01
101 RCL 51
102 ST/ 56
103 ST/ 57
104 RCL 55
105 ST+ 56
106 RCL 54
107 ST+ 57
108 2
109 RCL 51
110 *
111 ST/ 56
112 ST/ 57
113 LASTX
114 RCL 50
115 /
116 ATAN
117 2
118 RCL 63
119 1
120 -
121 Y↑X
122 RCL 62
123 *
124 1
125 RCL 50
126 SIGN
127 -
128 2
129 /
130 +
131 90
132 *
133 +
134 D-R
135 ST* 56
136 ST* 57
137 RCL 60
138 RCL 50
139 /
140 RCL 61
141 +
142 2
143 /
144 RCL 35
145 *
146 ST+ 57
147 LASTX
```

```
148 RCL 06
149 /
150 RCL 65
151 *
152 STO 50
153 1
154 RCL 65
155 RCL 35
156 *
157 -
158 SQRT
159 ST/ 50
160 ST* 06
161 RCL 33
162 RCL 34
163 /
164 RCL 56
165 *
166 ST+ 50
167 LASTX
168 RCL 57
169 ST- 50
170 ST- 06
171 -
172 RCL 31
173 ST/ 50
174 /
175 RCL 34
176 SQRT
177 ST* 50
178 ST/ 06
179 /
180 STO 51
181 RCL 33
182 ST/ 50
183 ST/ 06
184 .END.
```

HP41C TRIAXIAL ELLIPSOID PROGRAM MAGNETICS C1

```
01*LBL "C1"
02 RCL 36
03 RCL 46
04 *
05 STO 52
06 LASTX
07 RCL 37
08 *
09 STO 38
10 RCL 38
11 RCL 46
12 *
13 STO 49
14 RCL 42
15 RCL 13
16 *
17 ST- 52
18 LASTX
19 RCL 43
20 *
21 ST- 38
22 RCL 13
23 RCL 44
24 *
25 ST- 49
26 RCL 17
27 RCL 18
28 *
29 RCL 19
30 *
31 X↑2
32 STO 51
33 STO 06
34 RCL 17
35 X↑2
36 STO 53
37 /
38 STO 50
39 RCL 52
40 X↑2
41 *
42 ST- 51
43 RCL 06
44 RCL 18
45 X↑2
46 ST+ 53
47 /
48 ST+ 50
49 RCL 38
50 X↑2
51 *
52 ST- 51
53 RCL 06
54 RCL 19
55 X↑2
56 ST+ 53
57 /
58 ST+ 50
59 RCL 49
60 X↑2
61 *
62 ST- 51
63 RCL 17
64 RCL 18
65 R-P
66 RCL 49
67 *
68 X↑2
69 ST- 50
70 RCL 18
71 RCL 19
72 R-P
73 RCL 52
74 *
75 X↑2
76 ST- 50
77 RCL 17
78 RCL 19
79 R-P
80 RCL 38
81 *
82 X↑2
83 ST- 50
84 RCL 46
85 RCL 13
86 R-P
87 X↑2
88 ST- 53
89 RCL 53
90 RCL 50
91 *
92 6
93 /
94 RCL 51
95 2
96 /
97 -
98 RCL 53
99 3
100 Y↑X
101 27
102 /
103 -
104 RCL 53
105 X↑2
106 RCL 50
107 3
108 *
109 -
110 9
111 /
112 SQRT
113 STO 51
114 3
115 Y↑X
116 /
117 ACOS
118 3
119 /
120 COS
121 RCL 51
122 *
123 2
124 *
125 RCL 53
126 3
127 /
128 -
129 STO 06
130 RCL 17
131 X↑2
132 +
133 ST/ 52
134 RCL 06
135 RCL 18
136 X↑2
137 +
138 ST/ 38
139 *
140 RCL 19
141 X↑2
142 RCL 06
143 +
144 ST/ 49
145 *
146 SQRT
147 RCL 52
148 RCL 19
149 *
150 RCL 11
151 RCL 38
152 *
153 +
154 RCL 12
155 RCL 49
156 *
157 +
158 /
159 RCL 52
160 RCL 38
161 RCL 49
162 R-P
163 X<>Y
164 RDN
165 R-P
166 X↑2
167 X<>Y
168 RDN
169 *
170 ST/ 52
171 ST/ 38
172 ST/ 49
173 RCL 34
174 RCL 06
175 RCL 17
176 X↑2
177 *
178 /
179 STO 65
180 SQRT
181 ASIN
182 TAN
183 STO 05
184 STO 60
185 .END.
```

HP41C TRIAXIAL ELLIPSOID PROGRAM MAGNETICS C2

```
01*LBL "C2"
02 RCL 10
03 RCL 51
04 *
05 ST- 52
06 RCL 11
07 RCL 50
08 *
09 ST+ 38
10 RCL 12
11 RCL 06
12 *
13 ST- 49
14 RCL 52
15 RCL 36
16 *
17 RCL 38
18 RCL 37
19 *
20 +
21 RCL 49
22 RCL 38
23 *
24 +
25 RCL 29
26 COS
27 *
28 RCL 52
29 RCL 39
30 *
31 RCL 38
32 RCL 40
33 *
34 +
35 RCL 49
36 RCL 41
37 *
38 +
39 RCL 29
40 SIN
41 *
42 +
43 RCL 04
44 COS
45 *
46 RCL 52
47 RCL 42
48 *
49 RCL 38
50 RCL 43
51 *
52 +
53 RCL 49
54 RCL 44
55 *
56 +
57 STO 52
58 RCL 04
59 SIN
60 *
61 +
62 RCL 32
63 ST* 52
64 *
65 RTN
66*LBL "GG"
67 RCL 10
68 RCL 36
69 *
70 RCL 11
71 RCL 37
72 *
73 +
74 RCL 12
75 RCL 38
76 *
77 +
78 STO 00
79 RCL 10
80 RCL 42
81 *
82 RCL 11
83 RCL 43
84 *
85 +
86 RCL 12
87 RCL 44
88 *
89 +
90 RCL 10
91 RCL 39
92 *
93 RCL 11
94 RCL 40
95 *
96 +
97 RCL 12
98 RCL 41
99 *
100 +
101 RCL 00
102 R-P
103 X<>Y
104 RDN
105 R-P
106 "J RES="
107 ARCL X
108 PRA
109 "I RES="
110 ARCL Y
111 PRA
112 R↑
113 RCL 05
114 RCL 29
115 -
116 X<>Y
117 +
118 "D RES="
119 ARCL X
120 PRA
121 ADV
122 LASTX
123 "B RES="
124 ARCL X
125 PRA
126 ADV
127 END
```

HP41C TRIAXIAL ELLIPSOID WORKED EXAMPLE (a) INDUCTION (refer to Figs)

```
F=[illegible]
I=-65.0
D=10.0

DEPTH=300.0
PLUNGE=45.0
AZIMUTH=230.0
TILT=-45.0
a=250.0
b=150.0
c=100.0
VOL=15,707,963.3

Ka=0.001000
D=90.0
I=0.0

Kb=0.001000
D=180.0
I=0.0

Kc=0.001000
D=0.0
I=90.0

BEARING=-80.0

N1=2.1036
N2=4.0715
N3=6.3912

X=-200.0
BZ=-11.4
BT=5.9

X=-100.0
BZ=-32.2
BT=23.6

X=0.0
BZ=-66.7
BT=57.1

X=100.0
BZ=-72.0
BT=67.5

X=200.0
BZ=-29.7
BT=28.3

J RES=59.8
I RES=-65.0
D RES=10.2

B RES=-79.8
```